

UDG01

Ultrasonic Depth Gauge

OPERATOR'S MANUAL

REVISION DATE 02/1993 -T

COPYRIGHT © 1992, CAMPBELL SCIENTIFIC CANADA CORP.

WARRANTY AND ASSISTANCE

The **UDG01 ULTRASONIC DEPTH GAUGE** is warranted by CAMPBELL SCIENTIFIC (CANADA) CORP. ("CSC") to be free from defects in materials and workmanship under normal use and service for **twelve (12) months** from date of shipment unless specified otherwise. CSC's obligation under this warranty is limited to repairing or replacing (at CSC's option) defective products. The customer shall assume all costs of removing, reinstalling, and shipping defective products to CSC. CSC will return such products by surface carrier prepaid. This warranty shall not apply to any CSC products which have been subjected to modification, misuse, neglect, accidents of nature, or shipping damage. This warranty is in lieu of all other warranties, expressed or implied, including warranties of merchantability or fitness for a particular purpose. CSC is not liable for special, indirect, incidental, or consequential damages.

Products may not be returned without prior authorization. To obtain a Return Merchandise Authorization (RMA), contact CAMPBELL SCIENTIFIC (CANADA) CORP., at (780) 454-2505. An RMA number will be issued in order to facilitate Repair Personnel in identifying an instrument upon arrival. Please write this number clearly on the outside of the shipping container. Include description of symptoms and all pertinent details.

CAMPBELL SCIENTIFIC (CANADA) CORP. does not accept collect calls.

Non-warranty products returned for repair should be accompanied by a purchase order to cover repair costs.



CAMPBELL SCIENTIFIC

C A N A D A C O R P .

11564 - 149 street - edmonton - alberta - T5M 1W7
tel 780.454.2505 fax 780.454.2655

www.campbellsci.ca


UDG01 DESIGN CHANGES

Recently the UDG01 has undergone some design changes. These changes are implemented on all sensors shipped after December 1, 1992. The serial numbers of the revised version are C1397 or greater.

The most prevalent change is the new base for the sensor which provides improved humidity protection for the electronics while allowing the required ventilation for the ultrasonic transducer. The transducer must have equalized pressure on both sides of the diaphragm to ensure proper operation. The new base is available as a retrofit kit for clients who purchased the original version and wish to improve humidity protection. It is strongly recommended for those clients operating in humid environments or making water level measurements.


The other notable change is due to the discontinuation of the Texas Instrument sonar module which has been replaced with the Polaroid version. The Polaroid module has an offset of -39 mm instead of -16 mm. Sensors with serial numbers greater than C1397 have the -39 mm offset indicated on the serial number label.

To ease programming of the UDG01 a single instruction is available to operate the sensor. The P108 instruction is currently available only on the CR10 datalogger.




PREVIOUS VERSION

CURRENT VERSION



EXPANDED VIEW OF NEW BASE

1. SPECIFICATIONS



CURRENT VERSION

POWER REQUIREMENTS:	+12 VOLTS D.C.
POWER CONSUMPTION:	200 μ A (Quiescent) 65 mA (Active for 0.2 sec.)
OUTPUT:	16 bit binary coded message
MEASUREMENT RANGE:	0.6 to 10 Meters
ACCURACY:	± 1 cm or 0.4% of Distance to Target (whichever is greatest)
RESOLUTION:	0.5mm
BEAM ACCEPTANCE ANGLE:	Approx. 20°
OPERATING TEMPERATURE:	-25 TO +50°C Standard Consult factory for extended temperature ranges
OPERATING HUMIDITY:	5 TO 95% non-condensing
MAXIMUM CABLE LENGTH:	1000 feet
DIMENSIONS:	Length 23 cm Diameter 7.3 cm
WEIGHT:	1 Kg

2. INTRODUCTION

The UDG01 depth sensor measures the distance from the sensor to a surface. The most common applications are monitoring snow depth or water level. The primary components of this depth sensor are the Polaroid Ultrasonic transducer and the Polaroid 6500 Sonar Ranging Module. Both components are specified to -25°C although they can be tested to -40°C. Together they form a means of "ranging" in a target by bouncing an ultrasonic wave, or series of pulses off the surface and listening for the return echo. The time from transmit to the return of the echo is the basis for obtaining the distance. Since the speed of sound in air varies with air temperature, an independent temperature measurement is required. In order to measure the UDG01, the data acquisition system used must have the following features:

1. Measure air temperature and store the temperature data in a memory location. This information is required so a temperature correction can be applied to the distance measurement.
2. Have at least 3 control ports which can act as both an input or output. Alternatively, if the control ports cannot accept inputs, then an analog channel can be used. The analog input can be single-ended with a 0 to 5 volt range and have the capability of being selectively activated in order to read the serial output train.

Goodison et al.(1988a) described factors which can affect the accuracy of measurement when used in snow monitoring. They discussed the effect of the nature of the snow surface, of blowing or falling snow at the time of the measurement and the effect of the air temperature through which the acoustic waves travel causing errors in the measurement. The nature of the snow surface is almost impossible to assess. Scattering of the acoustical signal as a result of blowing or falling snow is a problem which is difficult to solve. Invalid data are easily identified and interpolation can be made on relatively short periods.

An analysis was done in order to assess measurement errors due to the temperature distribution and vertical wind that exist between the sensor and measurement surface. A simple relation describing the air temperature distribution was used. This relation takes into account the stability of the air layer and uses air temperature and horizontal velocity at a certain height, surface temperature

and roughness. The values used were representative of the normal atmospheric conditions. At values for the extremes of these conditions, a deviation of less than 0.1% was determined for an unstable atmosphere and of less than 0.05% for a stable atmosphere. A similar analysis was done to review the influence of vertical wind on the measurement. Under very strong vertical wind speeds (5 m s^{-1}) the potential deviation due to vertical wind was less than 0.03%.

3. OPERATION

The UDG01 was designed to be a simple device to use and it is not necessary to understand the technical aspects of the sensor operation. However, this information is provided for those who require a more detailed description about the operating characteristics of this sensor.

The UDG01 is a sealed unit and should not be opened unless the desiccant is being replaced or if more than one sensor is used on a single datalogger. If two or more sensors are being used, the UDG01 must be opened to select the proper address identification (see Section 4 on Address Jumper Identification).

If the sensor is opened contrary to the description in Section 4 (Address Jumper Identification), the sensor can be damaged and the VOID label will be damaged which cancels the warranty.

4. ADDRESS JUMPER IDENTIFICATION

The UDG01 is sent from the factory configured with address 00. The UDG01 can have up to 16 addresses. If more than one UDG01 is used on a single datalogger, the address jumpers must be changed.

The UDG01 should only be opened in an office, laboratory or a similar dry environment. There is only one recommended way to open the UDG01 to access the jumpers. Always ensure that the UDG01 is disconnected from the datalogger or not in operation (remove the scan rate from the datalogger) prior to opening the sensor. While the sensor is open, it is a good time to verify and/or replace the desiccant pack.

With a Phillips screwdriver, remove the two screws that hold the ultrasonic transducer in place at the end of the cylinder. Carefully pull out the assembly which remains attached to the circuit board. The wires are long enough to allow the transducer and its enclosure to simply hang down out of the way. At the end of the circuit board, on the right hand side facing the cylinder end with the component side of the circuit board facing down, will be two sets of jumpers (see Figure 1A). The jumper set on the left represents the most significant address bit while the set on the right, the least significant bit. The 16 possible addresses, the jumper configuration, the binary code (two groups of two), and the control port pulse sequence are all shown in Figure 1B.

If the signal wires are removed from the clips, please ensure that the polarity is correct. The wire with the red heat shrink or the solid red wire for S/N 1396 or smaller is attached to the tab that is connected to the red wire.

5. OPERATING THE UDG01 WITH A CR10


5.1 CONNECTIONS TO CR10

CAUTION: The order in which connections are made is critical. Always connect ground first, followed by 12V and then the Control Ports. In disconnecting the sensor the reverse order should be followed. The clear lead should be connected to the ground (shield). The UDG01 sensor needs to be attached to the cable via the environmental connector prior to wiring the cable end to the datalogger. The cable should be disconnected from the datalogger

prior to removing the UDG01 from operation. The 1K resistor enclosed, is required when using a 21X Micrologger only.

WIRE COLOR	FUNCTION
OR SINGLE BLACK BLACK w/RED	GROUND
RED	+12 VOLT
BLACK w/WHITE . . . OR YELLOW	CONTROL PORT 3 SDE (SERIAL DEVICE ENABLE)
WHITE	CONTROL PORT 2 SC CLOCK
GREEN	CONTROL PORT 1 DATA

WARNING! DO NOT OPERATE THE SENSOR WITHOUT THE POLAROID TRANSDUCER BEING ATTACHED OR DAMAGE COULD OCCUR.




CR10 WIRING DIAGRAM

5.2 P108 Instruction Programming (CR10 Only)

The P108 instruction is available for the CR10 datalogger. When using P108 the UDG01 can only be used with control Ports 1, 2 and 3 as indicated in the above wiring list. If you are not sure whether or not your CR10 contains the P108 instruction, try entering the instruction. Newer CR10s will generate Error E40 indicating the

U0G01 CROSS SECTION, END VIEW.



NOTE: Address Jumpers Shown with Factory Setting

FIGURE 1A. Address Jumper Location

ADDRESS	JUMPER CONFIGURATION		BINARY CODE		CONTROL PORT 1 PULSE SEQUENCE	
	0 1 2 3	0 1 2 3	0 - 3	0 - 3		
00			00	00		
01			00	01		
02			00	10		
03			00	11		
10			01	00		
11			01	01		
12			01	10		
13			01	11		
20			10	00		
21			10	01		
22			10	10		
23			10	11		
30			11	00		
31			11	01		
32			11	10		
33			11	11		

FIGURE 1B. Address Jumper Identification

instruction does not exist. Older CR10s will not allow you to enter an Instruction which does not exist. If the P108 instruction is not in your datalogger, refer to the sample program in section 5.4.

P108

Parameter	Data Type	Description
01:	2 ¹	UDG01 Address (00-33 Base 4)
02:	4--	Air Temperature (°C) Location
03:	4--	Destination Location
04:	FP	Multiplier
05:	FP	Offset

Note¹: Refer to Figure 1B for a valid address

5.3 SAMPLE P108 PROGRAM

The following program uses the P108 instruction to obtain a reading from the UDG01 with address 00 (Base 4). The UDG01 is shipped from the factory with this address.

*1 Table 1 Programs

01: 0.0000 Sec. Execution Interval, User selectable

The P11 instruction is used to obtain an air temperature measurement to compensate for the changes in the speed of sound in varying air temperatures.

01: P11 Temp 107 Probe
 01: 1 Rep
 02: 1 In Chan
 03: 1 Excite all reps w/EXchan 1
 04: 1 Loc [T1 Deg C]
 05: 1 Mult
 06: 0 Offset

In the P108 instruction listed below, the offset of -39 mm applies to UDG01 sensors with Polaroid sonic boards. This offset which is written on the serial number label applies to sensors with S/N of C1397 or greater. If the offset is not indicated on the serial number label, then use -16 mm as the offset value.

02: P108 UDG01 Sensor
 01: 00 UDG01 address (00-33 Base 4)
 02: 1 Air Temperature loc T1 Deg C
 03: 2 Dest. Loc[Distance]
 04: 1 Mult
 05: -39 Offset (Applicable Offset)

03: P End Table 1

To convert the distance in units of inches instead of millimeters the following multiplier and offset would be used.

02: P108 UDG01 Sensor
 01: 00 UDG01 address (00-33 Base 4)
 02: 1 Air Temperature loc T1 Deg C
 03: 2 Dest. Loc[Distance]
 04: .03937 Mult
 05: -1.535 Offset (Applicable Offset)

5.4 SAMPLE PROGRAM CR10 (NO P108)

This sample program is written for the CR10. It is an example only and the concepts illustrated here are likely to be only fragments of a larger program. This program is used to make distance measurements using the UDG01 Ultrasonic Depth Sensor. An address must first be sent to select 1 of up to 16 depth sensors. After the sensor is selected, a measurement is made and the result "shifted" to the control port of the datalogger.

This entire program is available in the P108 instruction (see Section 5.3) of some CR10 system PROMs. You need not review this program if your CR10 has the P108 instruction.

*1 Table 1 Programs

01: 0.0000 Sec. Execution Interval, User selectable

The following instruction measures air temperature and converts the measurement from degrees Celsius to degrees Kelvin. The result will be used later in the program to compensate for changes in the speed of sound due to temperature. Please note that in this example, a 107 temperature probes is used. Another type of air temperature sensor may be used. The important thing to remember is the location where the temperature information is being stored.

01: P11	Temp 107 Probe
01: 1	Rep
02: 1	In Chan
03: 1	Excite all reps w/EXchan 1
04: 1	Loc [T1 Deg K]
05: 1	Mult
06: 273.15	Offset

The following set of instructions configures the three control ports and addresses the UDG01 (the address used in this example is "0000"). Control Port 3, the Serial Device Enable line is used to control the sensor. At first it is set high to reset the UDG01. Control Port 2, the Clock line, is pulsed to receive each data bit from Control Port 1, the data line. Control Port 1 is used to first send 1,0,0,0, to the UDG01. The 1 will be used by the UDG01 to prevent any other data from entering before its time while the three 0's are for future potential use. Control Port 1 then sends the UDG01 address which in this case is 0000. Remember that for every data bit sent from Control Port 1, Control Port 2 is pulsed by the datalogger to receive the data. The first time Control Port 2 is pulsed is to make certain that the UDG was reset. The next seven pulses shift the first seven bits of the binary message. The eighth bit of the message is read when Control Port 2 is set high.

Figure 2 is a diagrammatical representation of the control port changes which occur up until the actual reception of the data from the UDG01.

02: P20	Set Port(s)
01: 9999	C8..C5=nc/nc/nc/nc
02: 9130	C4..C1=nc/high/1ms/low
03: 86	Do
01: 72	Pulse Port 2
04: P86	Do
01: 41	Set high Port 1
05: P86	Do
01: 72	Pulse Port 2
06: P86	Do
01: 51	Set low Port 1

In this example, the binary code 1,0,0,0,0,0,0,0 needs to be sent to the UDG01. In the above instructions the 1 was sent to the sensor and the data line was set low. Now that the data line is low all that is required is to clock Control Port 2 six times which tells the UDG01 to read the value at Control Port 1 (which is now 0) at each rising edge of the clock. To accomplish this a Loop is created. All this Loop is doing is pulsing Control Port 2 which clocks in six 0's to UDG01 from Control Port 1. At present, the only change in the message will be the last four bits, the address of the UDG01. If only one UDG01 is used, the program will not require changes.

With two or more sensors, the message must be changed since the address of the devices will be different. For example if the address of the UDG01 is 01, the binary string will be 1,0,0,0,1,0,0,0. This means that the Loop instruction should be changed to 3 Loop Counts instead of the 6 shown. Following this, will be the instruction to Set Port 1 High, then Pulse Port 2 to receive that bit, followed by Set Port 1 Low. Pulsing Port 2 twice and then Setting Port 2 High ends this sequence. As mentioned earlier, the 8th bit is read when Port 2 is set High.

Another example is if the address of the sensor was 03, then the binary string on Control Port 1 would be 1,0,0,0,1,1,0,0. This would mean that the Loop Count would have to be 3 followed by Set Port 1 High and Pulse Port 2 twice. Set Port 1 low, Pulse Port 2 and finally Set Port 2 High.

07: P87	Beginning of Loop
01: 0	Delay
02: 6	Loop Count
08: P86	Do
01: 72	Pulse Port 2
09: P95	END
10: P86	DO
01: 42	Set high Port 2

At this point in the program, the UDG01 has been properly addressed and is now ready to perform the measurement. If several UDG01's are used, the balance of the program can be operated with a subroutine called from this point in the program. Remember that subroutines must be placed in Table 3 (*3) of the datalogger. Since there may be several sensors used

and therefore calling the subroutine several times, data from each UDG01 must be moved into separate locations to avoid overwrites. This can be accomplished by using Instruction 31 "n" times, where n is the number of UDG01's on the same datalogger. Instruction 31 simply takes data from one location and stores it in another.

The following instruction creates a 10ms delay from the time that the UDG01 is selected and prior to the measurement being made. The delay allows the circuitry time to power up after it has been selected. Note that the excitation channel is not connected since this instruction is used to create a delay and not excite a device.

11: P22	Excitation with Delay
01: 1	EX Chan
02: 1	Delay w/EX (units = 0.01 sec)
03: 0	Delay after EX (units = 0.01 sec)
04: 0	mV Excitation

The next two instructions initiate the measurement.

12: P86	Do
01: 53	Set low Port 3
13: P86	DO
01: 43	Set high Port 3

The following instruction creates an 80ms delay. This is necessary for the sound wave to have enough time to travel a minimum of 10 meters and be reflected back before continuing in the program.

14: P22	Excitation with Delay
01: 1	EX Chan
02: 8	Delay w/EX (units = 0.01 sec)
03: 0000	Delay after EX (units = 0.01 sec)
04: 0.0000	mV Excitation
15: P20	Set Port(s)
01: 9999	C8..C5=nc/nc/nc/nc
02: 9908	C4..C1=nc/nc/low/input

The next two instructions set location 3 to a value of 1, and location 4 to a value of 0. Location 3 contains the exponent for the mantissa 2. Location 4 will contain the unprocessed pulse count.

16: P30	Z=F
01: 1	F
02: 0	Exponent of 10
03: 3	Z Loc[2 exp n]
17: P30	Z=F
01: 0	F
02: 0	Exponent of 10
03: 4	Z Loc[# pulses]

The following instructions "shift" the encoded pulse total to the datalogger, decodes the total and stores the total in location 4.

18: P87	Beginning of Loop
01: 0	Delay
02: 16	Loop Count
19: P86	Do
01: 72	Pulse Port 2
20: P91	If Flag/Port
01: 41	Do if port 1 is high
02: 30	Then Do
21: P33	Z=X+Y
01: 3	X Loc 2 exp n
02: 4	Y Loc # pulses
03: 4	Z Loc [# pulses]
22: P95	End
23: P37	Z=X*F
01: 3	X Loc 2 exp n
02: 2	F
03: 3	Z Loc [2 exp n]
24: P95	End

The following instructions are used to temperature compensate the measurement and convert the result into "distance to target". The result is stored in millimeters and can be found in location 7. The equation used is: Distance = SQRT $\left[\left(T_1/T_2\right)^{0.52176} \cdot N\right]$, where T_1 is the present temperature, T_2 is the fixed reference temperature of 18°C (291°K), 0.52176 is a constant to obtain the output in mm and N is the number of pulses (a 16 bit binary number read from the UDG01) generated by the crystal while the sound wave travelled to and from the surface.

25: P30	Z=F
01: 291	F
02: 0	Exponent of 10
03: 2	Z Loc [T2 (291K)]
26: P38	Z=X/Y
01: 1	X Loc T1 Deg K
02: 2	Y Loc T2 (291K)
03: 5	Z Loc [T1/T2]
27: P39	Z=SQRT(X)
01: 5	X Loc T1/T2
02: 6	Z Loc [sqrtT1/T2]
28: P36	Z=X*Y
01: 4	X Loc # pulses
02: 6	Y Loc sqrt T1/T2
03: 7	Z Loc [Result]
29: P37	Z=X*F
01: 7	X Loc Result
02: .52176	F
03: 7	Z Loc [Result]

This next instruction removes a built in offset of 39 mm from the calculated distance. Please note that some UDG01 sensors have an offset of 16 mm instead of 39 mm which needs to be removed (refer to the UDG01 design changes at the beginning of this manual).

30: P34	Z=X+F
01: 7	X Loc Result
02: -39	F (Applicable offset)
03: 7	Z Loc [Result]

The measurement is now complete and the last instruction returns the UDG to the quiescent mode.

```

31: P20      Set Port(s)
01: 9999     C8..C5=nc/nc/nc/nc
02: 9000     C4..C1=nc/low/low/low

32: P        End Table 1

```

6. MONITORING THE UDG01 WITH A 21X

Program: The example program is written for the 21X. It is an example only and the concepts illustrated here are likely to be only fragments of a larger program. This

program is used to make distance measurements using the UDG01 Ultrasonic Depth Sensor. An address must first be sent to select 1 of up to 16 depth sensors. After the sensor is selected, a measurement is made and the result "shifted" to the input channel of the datalogger. The program can be made into a subroutine and called when required.


6.1 CONNECTIONS TO 21X

CAUTION: The order in which the connections are made is critical. Always connect ground first, followed by +12V and then the Control Ports. In disconnecting the sensor the reverse order should be followed. The clear lead should be connected to the ground (shield). The UDG01 sensor needs to be attached to the cable via the environmental connector prior to wiring the cable end to the datalogger. The cable should also be disconnected from the datalogger prior to removing the UDG01 from operation.

```

OR SINGLE BLACK
BLACK w/RED ..... GROUND
RED ..... +12 VOLT
BLACK w/WHITE ... CONTROL PORT 3 SDE
OR YELLOW          (SERIAL DEVICE
                    ENABLE)
WHITE ..... CONTROL PORT 2
                CLOCK
GREEN ..... SE ANALOG CHANNEL
                (2 in this example)

```



WIRING DIAGRAM FOR 21X

A 1K resistor is connected between Control Port 1 and the Single-Ended Analog Channel. The resistor is incorporated in a separate light green lead that was sent with the UDG01.

6.2 SAMPLE PROGRAM 21X

*1 Table 1 Programs

01:0.0000 Sec Execution Interval

The following instruction measures air temperature and converts the measurement from degrees Celsius to degrees Kelvin. The result will be used later in the program to compensate for changes in the speed of sound due to temperature. Please note that in this example, a 107 temperature probes is used. Another type of air temperature sensor may be used. The important thing to remember is the location where the temperature information is being stored.

01: P11	Temp 107 Probe
01: 1	Rep
02: 1	In Chan
03: 1	Excite all reps w/EXchan 1
04: 1	Loc [T1 Deg K]
05: 1	Mult
06: 273.15	Offset

The following set of instructions configures the three control ports and addresses the UDG01 (the address used in this example is "0000"). Control Port 3, the serial device enable line, is used to control the sensor. At first it is set high to reset the UDG01. Control Port 2, the Clock line, is pulsed to receive each and every data bit from Control Port 1, the data line. Control Port 1 is used to at first send 1,0,0,0, to the UDG01. The 1 will be latched through the UDG01 to prevent any other data from entering before its time while the three 0's are for future potential use. Control Port 1 then sends the UDG01 address which in this case is 0000. Remember that for every data bit sent from Control Port 1, Control Port 2 is pulsed to receive the data. The first time Control Port 2 is pulsed is to make certain that the UDG01 was reset. The next seven pulses shift the first seven bits of the message, the eighth bit is read when Port 2 is set high (Refer to Figure 2 for a diagrammatical representation).

02: P20	Set Port
01: 1	Set high
02: 3	Port Number

03: P20	Set Port
01: 0	Set low
02: 1	Port Number

04: P20	Set Port
01: 1	Set high
02: 2	Port Number

05: P20	Set Port
01: 0	Set low
02: 2	Port Number

06: P20	Set Port
01: 1	Set high
02: 1	Port number

07: P20	Set Port
01: 1	Set high
02: 2	Port Number

08: P20	Set Port
01: 0	Set low
02: 2	Port Number

09: P20	Set Port
01: 0	Set low
02: 1	Port Number

In this example, the binary code 1,0,0,0,0,0,0 needs to be sent to the UDG01. In the above instructions the 1 was sent to the sensor and the data line was set low. Now that the data line is low all that is required is to clock Control Port 2 six times which tells the UDG01 to read the value at Control Port 1 (which is 0) at each rising clock edge. To accomplish this a Loop is created. All this Loop is doing is pulsing Control Port 2 which clocks in six 0's to UDG01 from Control Port 1. At present, the only thing that will change in the message are the last four bits, the address of the UDG01. If only one UDG01 is used, the program will not require changes.

With two or more sensors, the message must be changed since the address of the devices will be different. For example, if the address of the UDG01 is 01, the binary string will be 1,0,0,0,1,0,0,0. This means that the loop instruction should be changed to 3 Loop Counts instead of 6. Following this, will be the instruction to Set Port 1

High, then Pulse Port 2 to receive that bit and then Set Port 1 low. Port 2 must then be pulsed twice and then finally, Port 2 is set High. As mentioned earlier, the 8th bit is read when Port 2 is set High.

If the address of the sensor was 03, then the binary string on Control Port 1 would be 1,0,0,0,1,1,0,0. This would mean that the Loop Count would have to be 3 followed by Set Port 1 High and Pulse Port 2 twice, to receive these bits. Port 1 is then set low and Port 2 is pulsed once and finally Port 2 is set High.

10: P87	Beginning of Loop
01: 0	Delay
02: 6	Loop Count
11: P20	Set Port
01: 1	Set high
01: 2	Port Number
12: P20	Set Port
01: 0	Set low
02: 2	Port Number
13: P95	End
14: P20	Set Port
01: 1	Set high
02: 2	Port Number

At this point in the program, the UDG01 has been properly addressed and is ready to perform the measurement. If several UDG01's are used, the balance of the program can be operated with a subroutine called from this point in the program. A reminder that subroutines are placed in Table 3 (*3) of the datalogger. Since there may be several sensors used, and therefore calling the subroutine several times, data from each UDG01 must be moved into separate locations to avoid overwrites. This can be accomplished by using instruction 31 "n" times, where n is the number of UDG01's on the same datalogger. Instruction 31 takes data from one location and puts it in another.

The following instruction is used to create a 10ms delay between the time that the UDG is selected and prior to the measurement. This allows the circuitry time to power up after it has been selected. Note that there is no excitation channel connected since this instruction is used to create a delay and not excite any device.

15: P22	Excitation with Delay
01: 1	EX chan
02: 1	Delay w/EX (units = 0.01 sec)
03: 0	Delay after EX (units = 0.01 sec)
04: 0	mV Excitation

The next two instructions initiate the measurement.

16: P20	Set Port
01: 0	Set low
02: 3	Port Number
17: P20	Set Port
01: 1	Set high
02: 3	Port Number

The following instruction creates an 80ms delay. This is necessary for the sound wave to have enough time to travel a minimum of 10 meters and be reflected back before continuing in the program.

18: P22	Excitation with Delay
01: 1	EX chan
02: 8	Delay w/EX (units = 0.01 sec)
03: 0	Delay after EX (units = 0.01 sec)
04: 0	mV Excitation
19: P20	Set Port
01: 0	Set high Low
02: 2	Port Number

The next two instructions set location 3 to a value of 1, and location 4 to a value of 0. Location 3 contains the exponent for the mantissa 2. Location 4 will contain the unprocessed pulse count.

20: P30	Z=F
01: 1	F
02: 3	Z Loc [2 exp n]
21: P30	Z=F
01: 0	F
02: 4	Z Loc [# pulses]

The following instructions "shift" the encoded pulse total to the datalogger, decodes the total and stores the total in location 4.

22: P87	Beginning of Loop
01: 0	Delay
02: 16	Loop Count

23: P20	Set Port
01: 1	Set high
02: 2	Port Number
24: P20	Set Port
01: 0	Set low
02: 2	Port Number
25: P1	Volt (SE)
01: 1	Rep
02: 5	5000 mV slow Range
03: 2	In Chan
04: 8	Loc [HI OR LO]
05: 1	Mult
06: 0.0000	Offset
26: P89	If X<=>F
01: 8	X Loc HI OR LO
02: 3	>=
03: 3000	F
04: 30	Then Do
27: P33	Z=X+Y
01: 3	X Loc 2 exp n
02: 4	Y Loc # pulses
03: 4	Z Loc [# pulses]
28: P95	End
29: P37	Z=X*F
01: 3	X Loc 2 exp n
02: 2	F
03: 3	Z Loc [2 exp n]
30: P95	End

32: P38	Z=X/Y
01: 1	X Loc T1 Deg K
02: 2	Y Loc T2 (291K)
03: 5	Z Loc [T1/T2]
33: P39	Z=SQRT(X)
01: 5	X Loc T1/T2
02: 6	Z Loc[sqrtT1/T2]
34: P36	Z=X*Y
01: 4	X Loc # pulses
02: 6	Y Loc sqrtT1/T2
03: 7	Z Loc[Result]
35: P37	Z=X*F
01: 7	X Loc Result
02: .52176	F
03: 7	Z Loc [Result]
36: P34	Z=X+F
01: 7	X Loc Result
02: -39	F (Applicable offset)
03: 7	Z Loc [Result]

The measurement is now complete and the last 3 instructions return the UDG to the quiescent mode.

37: P20	Set Port
01: 0	Set low
02: 1	Port Number
38: P20	Set Port
01: 0	Set low
02: 2	Port Number
39: P20	Set Port
01: 0	Set low
02: 3	Port Number

The following instructions are used to temperature compensate the measurement and convert the result into "distance to target". The result is stored in millimeters and can be found in location 7. The equation used is: Distance = SQRT[(T₁/T₂)]*0.52176*N, where T₁ is the present temperature, T₂ is the fixed reference temperature of 18°C (291°K), 0.52176 is a constant to obtain the output in mm and N is the number of pulses (a 16 bit binary number read from the UDG01) generated by the crystal while the sound wave travelled to and from the surface.

31: P30	Z=F
01: 291	F
02: 2	Z Loc [T2 (291K)]

7. FIELD INSTALLATION

The UDG01 has two mounting options for either a tripod or a tower installation (see Figures 3A and 3B). The UDG01 small mounting elbow will screw into a 1/2 inch thread. Both mounting arms the 019UDGTOW and the 019UDGTRI have a common 1 1/4 in. pipe with a reduction to 1/2 in. to accept the UDG01. Once the sensor has been installed, the result will show the distance from the sensor to the surface. An independent measurement should be made to confirm the accuracy of the results. After this, the choice is whether to leave the results as simply the distance to target or to remove this initial distance as an offset. It will depend on the monitoring application. If snow accumulation is required, it is probably best to "zero" the initial distance and as the snow accumulates below the sensor, the measurement will simply show snow depth. To do this, the multiplier must be -1 and the offset would be positive.

For example, suppose that the sensor is installed at 5 meters or 5000 mm above the surface. In order to obtain accumulation, the offset would be +5039 while the multiplier would be -1. At first, the UDG01 would produce an initial uncorrected reading of +5039. By making the offset +5039 (incorporating the distance to surface with the 39 mm built in offset) and the multiplier -1., the results would be 0.0. Assume that 200 mm of snow falls. The sensor would normally read 4839 (4800, the distance to target + 39 mm built in offset). With the -1 multiplier and the offset of +5039, the final reading would be $-4839 + 5039 = 200$ mm. If the value to be output is greater than 6999, then high resolution data must be used. Also, if greater accuracy is required and to take advantage of the 0.5 mm resolution of the UDG01 high resolution data must be output.

If desired one may work with the distance to target and use that as a reference, then as the distance fluctuates up or down, as for example in a water level application, use the initial distance measurement to correct the data.




FIGURE 3A. Diagrammatic representation of UDG01 mounted to tower




FIGURE 3B. Diagrammatic representation of UDG01 mounted to CM10 Tripod Mast

DATA INTERPRETATION

Figure 4 is a graph from our previous depth sensor. This graph is shown to illustrate the type of signal obtained when monitoring snow depth. After installation at the end of May, the sensor measured an initial snowfall which then gradually compacted, and/or sublimated until the next event on June 14. There is a data gap after this which is as a result of the measurement being done during the snowfall event. Another missing data point is seen on June 28 and is either again as a result of snow falling during the time of measurement or blowing snow.

Figure 5 contains two graphs of a UDG01 measurement series on a fixed distance. Figure 5A is for the UDG01 at room temperature while Figure 5B is the UDG01 in a strongly varying temperature regime from -18°C to -60°C . Note the stability of the signal over this wide range of temperatures.

REFERENCES

Here is a short list of reference material which deals with the original acoustic sensor as it was developed by the Atmospheric Environment Service. Most of the technical details are different than those of the UDG01. This reference list is included to provide the user a better understanding of field conditions and data interpretation. These references also present graphed data which will assist in interpreting data when using the sensor in a snow depth monitoring application.

Goodison, B.E., J.R. Metcalfe, R.A. Wilson, 1988a. Performance of a Canadian Automatic snow depth sensor. Proc. WMO Technical Conference, TECO-88, Leipzig, German Democratic Republic, May 25 - June 1, 1988 (in press).




FIGURE 4. Daily snow depth from Agassiz Ice Cap, Ellesmere Is.




FIGURE 5A. Measurement of a fixed distance at room temperature.




FIGURE 5B. Measurement of a fixed distance under a wide temperature range. (-18°C to -60°C)